The NumericalCertification package in Macaulay2

Kisun Lee

Abstract

The package NumericalCertification implements methods for certifying numerical approximations of solutions for a given system of polynomial equations. For certifying regular solutions, the package implements Smale’s $\alpha$-theory and Krawczyk method. For a singular solution, we implement soft verification using the iterative deflation method. We demonstrate the functionalities of the package focusing on interaction with current numerical solvers in Macaulay2.

1 Introduction

Systems with polynomial equations arise in many fields in mathematics and applied science. Specially, the interest on polynomial systems turns into a problem in algebraic geometry as finding all isolated solutions for a given polynomial system. Due to a recent development in numerical algebraic geometry (e.g. see [SW05]), a family of numerical algorithms called the homotopy continuation gains popularity as a way to find solutions for a polynomial system. There are several known implementations Bertini [BHSW], HomotopyContinuation.jl [BT18], Hom4PS-3 [CLL14], NumericalAlgebraicGeometry [Ley11] and PHCpack [Ver99] which are all widely used.

One remark for the homotopy continuation algorithm is that its output is not certified. It means that numerical approximations obtained by the algorithm might not satisfy the users depending on their purposes.

We say that a numerical approximation is certified if a compact region that contains a unique solution can be obtained from the given approximation by applying a sort of algorithm. For a certified approximation, the unique solution contained in the compact region is called an associated solution and the given approximation is called an approximate solution. We call this series of algorithms numerical certification.

As implementations for numerical certification for a polynomial system, we point out alphaCertifed [HS11] and a function certify in the software HomotopyContinuation.jl [BRT20]. alphaCertifed implements Smale’s $\alpha$-theory [Sma86] as a way for numerical certification [HS12]. On the other hand, HomotopyContinuation.jl exploits Krawczyk method [Kra69] using interval arithmetic [Moo77] as a tool for certification.

The package NumericalCertification in Macaulay2 [GS02] executes regular solution certification using both $\alpha$-theory and Krawczyk method. A preferred method can be chosen as an option by users. As an improved version of the software presented in [Lee19], it includes soft verification for a singular solution using the idea of the deflation method [LVZ06]. Finally, the package provides an interface to the software alphaCertifed.

The rest of the paper consists of two sections. In the next section, we discuss the required preliminaries for certification. The implementation details are given in the last section.

2 Preliminaries

In this section, we review the concepts used for numerical certification. Smale’s $\alpha$-theory and a combination of Krawczyk method and interval arithmetic are used as methods for regular solution certification. We introduce the deflation method for a notion providing an idea for singular solution certification.
2.1 Smale’s α-theory

Consider an \( n \times n \)-square system \( F \), i.e. a system with \( n \) polynomial equations with \( n \) variables. For a point \( x \in \mathbb{C}^n \), recall the \textbf{Newton operator} \( N_F(x) \) defined like the following:

\[
N_F(x) = \begin{cases} 
  x - F'(x)^{-1}F(x) & \text{if } F'(x) \text{ is invertible}, \\
  x & \text{otherwise}
\end{cases}
\]

We say that a sequence \( \{N^k_F(x)\}_{k=1}^{\infty} \) converges quadratically to an associated solution \( x^* \) of \( F \) if for every \( k \in \mathbb{Z}_+ \),

\[
\|N^k_F(x) - x^*\| \leq \left( \frac{1}{2} \right)^{k-1} \|x - x^*\|.
\]

In this case, \( x \) is an approximate solution for \( F \). When \( F'(x) \) is not invertible, we say \( x \) is an approximate solution if and only if \( F(x) = 0 \). The \( \alpha \)-theory provides a certificate for the quadratic convergence of a given point. The certificate is obtained from the three auxiliary parameters:

\[
\begin{align*}
\alpha(F, x) &:= \beta(F, x) \gamma(F, x) \\
\beta(F, x) &:= \|x - N_F(x)\| = \|F'(x)^{-1}F(x)\| \\
\gamma(F, x) &:= \sup_{k \geq 2} \left\| \frac{F^{(k)}(x) - F^{(k)}(x)}{k!} \right\|
\end{align*}
\]

where \( F^{(k)}(x) \) in the definition of \( \gamma(F, x) \) is a symmetric tensor whose components are the \( k \)-th partial derivatives of \( F \), see [Lan83, Chapter 5]. The norm in \( \beta(F, x) \) is the usual Euclidean norm and the norm in \( \gamma(F, x) \) is the operator norm on \( S^k \mathbb{C}^n \) (for details, see [HS11]). When \( F' \) is not invertible at \( x \), we define \( \alpha(F, x) = \beta(F, x) = \gamma(F, x) = \infty \). The contents of \( \alpha \)-theory are summarized below.

\textbf{Theorem 2.1} (c.f. [BCSS12, HS12]). Let \( F : \mathbb{C}^n \to \mathbb{C}^n \) be a square polynomial system with a point \( x \in \mathbb{C}^n \). Then,

1. if \( \alpha(F, x) < \frac{13-3\sqrt{17}}{4} \), then \( x \) is an approximate solution for \( F \), and \( \|x - x^*\| \geq 2\beta(F, x) \) where \( x^* \) is an associated solution to \( x \).

2. if \( \alpha(F, x) < 0.03 \) and \( \|x - y\| < \frac{1}{2\beta(F, x)} \) for a point \( y \), then \( x \) and \( y \) are both approximate solutions for \( F \) to the same solution \( x^* \), and \( \|x - x^*\| \leq \frac{1}{2\gamma(F, x)} \).

3. if \( \|x - \overline{x}\| > 4\beta(F, x) \) for the conjugate \( \overline{x} \) of \( x \), then \( x^* \) is not real.

For an implementation of \( \alpha \)-theory, the step for computing (or bounding) \( \gamma \) is required. For a degree \( d \) polynomial \( f = \sum_{|\nu| \leq d} a_{\nu} x^\nu \), we recall that the \textbf{Bombieri-Weyl norm} is defined as

\[
\|f\|^2 = \frac{1}{d!} \sum_{|\nu| \leq d} \nu! (d - |\nu|)! |a_{\nu}|^2.
\]

For a system of polynomials \( F = \{f_1, \ldots, f_n\} \), we define a norm for the system

\[
\|F\|^2 = \sum_{i=1}^n \|f_i\|^2.
\]

Let \( d_i = \deg f_i \) for each \( i = 1, \ldots, n \) and \( d = \max d_i \). For a point \( x \in \mathbb{C} \), define \( \|(1, x)\|^2 = 1 + \sum_{i=1}^n |x_i|^2 \), and we let \( \Delta_F(x) \) be the diagonal matrix with entries \( \Delta_F(x)_{ii} := \sqrt{d_i} \|(1, x)\|^{d_i-1} \). Combining all these, a bound for \( \gamma(F, x) \) is given as follows:
Proposition 2.2. [HS12, Proposition 5] Let \( F \) be a square system of polynomials and \( x \in \mathbb{C}^n \) be a point. Suppose that \( F'(x) \) is nonsingular. Define

\[
\mu(F, x) := \max \left\{ 1, \| F \| \| F'(x)^{-1} \Delta_F(x) \| \right\}
\]

where the norm in \( \| F'(x)^{-1} \Delta_F(x) \| \) is the operator norm. Then,

\[
\gamma(F, x) \leq \frac{\mu(F, x)d^n}{2\| (1, x) \|}.
\]

2.2 Interval arithmetic and Krawczyk method

Interval arithmetic is introducing arithmetic operators between intervals to achieve conservative results on numerical computations. For an operator \( \odot \) with intervals \([a, b]\) and \([c, d]\) over \( \mathbb{R} \), we define \([a, b] \odot [c, d] = \{ x \odot y \mid x \in [a, b], y \in [c, d] \} \). The real interval arithmetic can be extended over \( \mathbb{C} \). For describing an interval over the complex, we use two intervals to construct an interval box \([a_1, b_1] + i[a_2, b_2]\) containing numbers in \( \mathbb{C} \). Then, complex interval arithmetic can be done similarly using arithmetic over the complex numbers. A set of complex intervals is denoted by \( \mathbb{I}\mathbb{C} \). Likewise, a set of \( n \)-dimensional complex interval boxes is denoted by \( \mathbb{I}^n \mathbb{C} \). For a function \( F : \mathbb{C}^n \rightarrow \mathbb{C}^n \) and an interval box \( I \in \mathbb{I}^n \mathbb{C} \), we define an interval extension of \( F \) as a set containing the image of all \( F \) on \( I \), and it is denoted by \( \square F(I) \).

Krawczyk method is a combination of interval arithmetic and generalized Newton’s method to get certificates for the existence and uniqueness of a solution for a square system in a given interval. Suppose that a square differentiable system \( F : \mathbb{C}^n \rightarrow \mathbb{C}^n \) is given with an interval \( I \subseteq \mathbb{I}^n \mathbb{C} \). Let \( Y \) be an \( n \times n \)-invertible matrix and \( x \) be a point in \( I \). Then, we define Krawczyk operator centered at \( x \) like the following:

\[
K_{x,Y}(I) := x - YF(x) + (I_n - Y\Box F'(I))(I - x)
\]

where \( I_n \) is the \( n \times n \) identity matrix. Then, the following theorem summarizes the propositions required for interval arithmetic-based certification.

Theorem 2.3 (c.f. [Kra69, BLL19, BRT20]). Suppose that \( F : \mathbb{C}^n \rightarrow \mathbb{C}^n \) is a square differentiable system with a given interval extension \( \Box F(I) \) on an interval \( I \). For an \( n \times n \)-invertible matrix \( Y \) and a point \( x \),

1. if a root \( x^* \) of \( F \) is in \( I \), then \( x^* \in K_{x,Y}(I) \).
2. if \( K_{x,Y} \subseteq I \), then \( I \) contains a root \( x^* \) of \( F \).
3. if \( I \) contains a root of \( F \) and \( \sqrt{2}\|I_n - Y\Box F'(I)\| < 1 \), then the root \( x^* \) in \( I \) is unique.
4. if \( I \) contains a root of \( F \), \( \sqrt{2}\|I_n - Y\Box F'(I)\| < 1 \) and a set of conjugates \( \{ \overline{y} \mid y \in K_{x,Y}(I) \} \) for \( K_{x,Y}(I) \) is contained in \( I \), then the root \( x^* \) in \( I \) is unique and real.

Here, \( \|I_n - Y\Box F'(I)\| \) is the maximum operator norm of the interval matrix \( I_n - Y\Box F'(I) \) under the max-norm.

Note that the invertible matrix \( Y \) is chosen for minimizing \( \|I_n - Y\Box F'(I)\| \). In an actual implementation, a natural choice for \( Y \) can be \( F'(m(I))^{-1} \) where \( m(I) \) is the midpoint of the box \( I \).

Remark 2.4. In general, interval arithmetic certification allows working with less precision than \( \alpha \)-theory. On the other hand, \( \alpha \)-theory shows a better convergence rate to an actual solution. An example in [BLL19, Section 5.1] shows a comparison between two methods.
2.3 The deflation method

A deflation is a series of methods to reinstate the quadratic convergence of Newton iteration for an isolated singular solution of a system of equations. The basic idea is introducing more equations to construct an augmented system with reduced singularity (e.g. multiplicity). For an isolated singular solution $x^*$ of a square system $F = \{f_1, \ldots, f_n\}$, define $\kappa = \dim \ker F'(x^*)$. Then, for a randomly chosen vector $B = [b_1, \ldots, b_n]^\top \in \mathbb{C}^{n \times 1}$ from the kernel of $F'(x^*)$, an augmented system

$$G = \begin{bmatrix} F \\ F' \times B \end{bmatrix}$$

has a solution $x^*$ with a lower multiplicity than that of $F$. It is known that a singular solution is regularized within finitely many iterations by applying iterative deflation [LVZ06].

3 Implementation details

The package `NumericalCertification` is designed to interplay with other numerical solvers in Macaulay2, for example, `NumericalAlgebraicGeometry` [Ley11], Bertini [BGLR13] or PHCpack [GPV11]. Hence, the package supports `PolySystem` and `AbstractPoint` types of input.

The most direct way to use the package is `certifySolutions`. It takes a polynomial system and a list of numerical solutions as input.

```plaintext
i1 : needsPackage "NumericalCertification"
i2 : R = CC[x1,x2,y1,y2];
i3 : F = polySystem {3*y1 + 2*y2 -1, 3*x1 + 2*x2 -3.5, x1^2 + y1^2-1, x2^2 + y2^2 -1};
i4 : sols = solveSystem f; -- a list of numerical solutions
i5 : c = certifySolutions(F,sols);
```

It supports three strategies as options, `alphaTheory`, `intervalArithmetic` and `alphaCertified`, and the default value is `alphaTheory`. The function returns `MutableHashTable` and we can peek the output to see the certification results. The option `alphaTheory` returns the values of $\alpha(F, x)$ for each numerical solution (in an order of the input), the list of certified distinct, real, regular and singular solutions. The list of non-certified solutions also returned and it may be certified again after refinement.

```plaintext
i6 : peek c
   o6 = MutableHashTable{alphaValues => {2.07811e-30, 1.97421e-40}
                   certifiedDistinct => {{.652548, .771177, .757747, -.63662},
                                         {.95437, .318445, -.298627, .947941}}
                   certifiedReal => {{.652548, .771177, .757747, -.63662},
                                        {.95437, .318445, -.298627, .947941}}
                   certifiedRegular => {{.652548, .771177, .757747, -.63662},
                                               {.95437, .318445, -.298627, .947941}}
                   certifiedSingular => {}
                   nonCertified => {} }
```

The option `intervalArithmetic` returns the list of Krawczyk operators for certified real, regular and singular solutions. For the list of non-certified solutions, it returns the input interval boxes used for certification.

```plaintext
i7 : c = certifySolutions(F,sols, Strategy=>"intervalArithmetic");
i8 : peek c
   o8 = MutableHashTable{certifiedReal => {[1.95437, 0.95437] +
                                                [-1.33962e-27, 1.33962e-27]*ii ... }
                   certifiedRegular => {[1.95437, 0.95437] +
                                           [-1.33962e-27, 1.33962e-27]*ii ... }
                   certifiedSingular => {}
                   nonCertified => {} }
```

The most direct way to use the package is `certifySolutions`. It takes a polynomial system and a list of numerical solutions as input.
For the option \texttt{alphaCertified}, we need to set a path to the software installed. It might require reloading the package. When we run \texttt{certifySolutions} with the option, it runs the software and creates files in the directory where \texttt{alphaCertified} is installed.

\begin{verbatim}
i9 : loadPackage("NumericalCertification",
    Configuration=>{"ALPHACERTIFIEDexec"=>"path/to/alphaCertified/"},
    Reload=>true)
i10 : certifySolutions(F,sols, Strategy=>"alphaCertified")
\end{verbatim}

\texttt{alphaCertified v1.3.0 (October 16, 2013)}
Jonathan D. Hauenstein and Frank Sottile
GMP v6.2.1 & MPFR v4.1.0

What follows is the implementation details of each strategy used in the package.

### 3.1 \(\alpha\)-theory certification

In this section, we look into functions executed in certification with \(\alpha\)-theory. These functions can also be used separately.

The function \texttt{computeConstants} takes a polynomial system \(F\) and a numerical point \(x\), and computes three parameter values \(\alpha(F,x), \beta(F,x)\) and \(\gamma(F,x)\). For the value of \(\gamma(F,x)\), the upper bound given in (2) is used instead. We use the Frobenius norm to bound the matrix norm used in \(\mu(F,x)\) (1). The polynomial system and its Jacobian are evaluated by a straight-line program (see [BCS13, Section 4.1]) implemented in the package \texttt{SLPexpressions} [CDLS] for a faster evaluation. Note that the polynomial system and the point must be in the same coefficient ring.

\begin{verbatim}
i11 : x = point{{.652548, .771177, .757747, -.63662_CC}};
i12 : computeConstants(F,x)
o12 = (1.16708e-10, 5.22384e-13, 223.414)
o12 : Sequence
\end{verbatim}

The function \texttt{certifyRegularSolution} certifies the given solution by checking the inequality \(\alpha(F,x) < \frac{13-3\sqrt{17}}{4}\). It returns true if input satisfies the inequality, false otherwise.

\begin{verbatim}
i13 : certifyRegularSolution(F,x)
o13 = true
\end{verbatim}

The function \texttt{certifyDistinctSolutions} takes a polynomial system and two points as input. It returns false if given two points converge to the same actual solution of the system, otherwise true.

\begin{verbatim}
i14 : y = point{{.95437, .318445, -.298627, .947941_CC}};
i15 : certifyDistinctSolutions(F,x,y)
o15 = true
\end{verbatim}

A given solution may converge to a solution over the real numbers even though it is a complex-valued solution. The function \texttt{certifyRealSolution} checks if a given solution corresponds to a real solution or not.
i16 : x = point{{.652548, .771177, .757747, -.63662+0.001*ii}};
i17 : certifyRealSolution(F,x)
o17 = true

The function alphaTheoryCertification takes a polynomial system and a list of numerical solutions and runs all aforementioned functions at once according to the algorithm established in [HS12, Section 2.2]. Unlike certifySolutions, it does not execute singular solution certification.

i18 : sols = {x,y};
i19 : c = alphaTheoryCertification(F,sols);
i20 : peek c
  o20 = MutableHashTable{alphaValues => {.000223414, 1.04693e-10}}
certifiedDistinct => {x, y}
certifiedReal => {x, y}
certifiedRegular => {x, y}

Finally, the package supports the exact arithmetic over the rational numbers or Gaussian rationals for α-theory certification. For example, constants α(F,x), β(F,x) and γ(F,x) can be computed over the Gaussian rationals as follows:

i21 : CR = QQ[i]/ideal(i^2+1); -- a ring of Gaussian rationals
i22 : R = CR[x1,x2,y1,y2];
i23 : F = polySystem {3*y1 + 2*y2 -1, 3*x1 + 2*x2 -7/2, x1^2 + y1^2 -1, x2^2 + y2^2 -1};
i24 : x = point(sub(matrix{{5/9,3/4,3/4,-1/2}},CR));
i25 : computeConstants(F,x)
o25 = (-----------------, ---------, -----------)
     8695980754208352 303595776 229146291
o25 : Sequence

3.2 Interval arithmetic certification

For interval arithmetic certification, the package provides a type of intervals over the complex numbers. The function intervalCCi returns a complex interval from a pair of real intervals representing real and imaginary part respectively.

i26 : I1 = intervalCCi(interval(.8,.9),interval(-0.1,0.1))
o26 = [.8,.9] + [-.1,.1]*ii
o26 : CCi

When only one real interval is given as input, it returns a complex interval with the zero interval for its imaginary part.

i27 : I2 = intervalCCi(interval(.2,.3))
o27 = [.2,.3] + [0,-0]*ii
o27 : CCi

The package supports a basic interval arithmetic for the complex intervals and matrices with complex interval entries.

i28 : I1 + I2
o28 = [1.1,1.2] + [-.1,.1]*ii
o28 : CCi
i29 : I1 * I2
o29 = [.16,.27] + [-.03,.03]*ii
The function `pointToInterval` constructs an interval box from a given point. This function helps to make interval input without defining them separately. There are two ways to use the function. The first is inputting a point and a desired radius for an interval box. Then, it returns an interval box with the given radius centered at the given point.

```plaintext
i33 : x = point{{-1.6,-1.3*ii}};
i34 : I = pointToInterval(x,1)
o34 = | [-2.6,-.6] + [-1,1]*ii | [-2.3,-.3] + [-1.27202,-1.27202]*ii |
o34 : CCiMatrix
```

In many cases, a proper radius can be different depending on the polynomial system or the accuracy of the approximation. Running the function `pointToInterval` with a polynomial system and a point as input returns an interval box with a radius estimating the distance between the point and the convergence limit by using Newton-Kantorovich theorem.

```plaintext
i35 : R = CC[x,y];
i36 : F = polySystem {x^2 + y^2 -1, x - y^2};
i37 : x = point{{-1.61803, -1.27202*ii}};
i38 : I = pointToInterval(F,x)
o38 = | [-1.61803,-1.61803] + [-2.2629e-25,2.2629e-25]*ii |
     | [-1.27202,-1.27202] + [-1.27202,-1.27202]*ii |
o38 : CCiMatrix
```

The function `krawczykOperator` computes Krawczyk operator from a given polynomial system and an interval box or a point. When a point is given as input, it computes Krawczyk operator from the interval obtained by `pointToInterval(F,x).

```plaintext
i39 : krawczykOperator(F,I)
o39 = | [-1.61803,-1.61803] + [-2.2629e-25,2.2629e-25]*ii |
     | [-1.65999e-25,1.65999e-25] + [-1.27202,-1.27202]*ii |
o39 : CCiMatrix
i40 : krawczykOperator(F,x)
o40 = | [-1.61803,-1.61803] + [-2.2629e-25,2.2629e-25]*ii |
     | [-1.65999e-25,1.65999e-25] + [-1.27202,-1.27202]*ii |
o40 : CCiMatrix
```

The function `krawczykTest` checks if Krawczyk operator satisfies 2 and 3 of Theorem 2.3. It returns true if it is (hence the given interval is certified to contain a solution uniquely), false otherwise.

```plaintext
i41 : krawczykTest(F,I)
o41 = true
```
Finally, the function `krawczykRealnessTest` certifies if a given interval corresponds to a real solution to the system or not. It returns true if the given interval contains a unique real solution to the system, false otherwise.

```plaintext
i42 : krawczykRealnessTest(F,I)
o42 = false
i43 : y = point{{.618034, -.786151}}; -- a real solution
i44 : krawczykRealnessTest(F,y)
o44 = true
```

As shown in `i44`, both functions `krawczykTest` and `krawczykRealnessTest` also take a point as input.

### 3.3 Singular solution certification

The method of iterated deflation is exploited for singular solution certification. For a polynomial system and a numerical approximation, we consider a subsystem

\[
\hat{F} = F + F'B = \left\{ f_i + \sum_{j=1}^{n} \frac{\partial f_i}{\partial x_j} b_j \right\}_{i=1,\ldots,n}
\]

of the overdetermined system \( G \) given in (3) obtained by the deflation. The \( \alpha \)-theory or interval arithmetic certification is applied on the square subsystem and the given numerical solution. If the numerical solution is still singular, then we construct an augmented system and take a square subsystem repeatedly. Since applying the deflation on a singular solution must terminate within finitely many iterations, the given numerical approximation becomes an approximation of a regular solution of the square subsystem.

It is possible to produce a false positive result as \( \hat{F} \) is obtained by squaring-up the overdetermined system from a randomly chosen vector \( B \). However, it can recover the quadratic convergence of Newton iteration for a singular solution with probability 1, and so it can be used as soft verification of a singular solution.

Singular solution certification is done by the function `certifySingularSolution` to a given polynomial system and a numerical solution.

```plaintext
i45 : F = polySystem {x^2 + y, x^3 - y^2};
i46 : x = point{{1e-7,2e-7*ii}};
i47 : certifySingularSolution(F,x)
o47 = true
```

Both strategies `alphaTheory` and `intervalArithmetic` are available as options. The function executes the iterated deflation until the given singular solution is regularized. Therefore, the function might not terminate if a poor approximation is given. To prevent this, if a user knows the number of iterations required in advance, it can be provided as input.

```plaintext
i48 : certifySingularSolution(F,x,1) -- an insufficient number of iterations
o48 = false
i49 : certifySingularSolution(F,x,2)
o49 = true
```

### Acknowledgements

The author is grateful to Anton Leykin for encouraging the project and suggesting various improvements. The author would also like to thank Michael Burr and Thomas Yahl for help on the package, and the organizers of the Macaulay2 workshop at Cleveland State University where the project was greatly improved.
References

[BCS13] Peter Bürgisser, Michael Clausen, and Mohammad A Shokrollahi. *Algebraic complexity theory*, volume 315. Springer Science & Business Media, 2013.

[BCSS12] Lenore Blum, Felipe Cucker, Michael Shub, and Steve Smale. *Complexity and real computation*. Springer Science & Business Media, 2012.

[BGLR13] Daniel J Bates, Elizabeth Gross, Anton Leykin, and Jose Israel Rodriguez. Bertini for macaulay2. *arXiv preprint arXiv:1310.3297*, 2013.

[BHSW] Daniel J. Bates, Jonathan D. Hauenstein, Andrew J. Sommese, and Charles W. Wampler. Bertini: Software for numerical algebraic geometry. Available at bertini.nd.edu with permanent doi: dx.doi.org/10.7274/R0H41PB5.

[BLL19] M. Burr, K. Lee, and A. Leykin. Effective certification of approximate solutions to systems of equations involving analytic functions. In *Proceedings of the 2019 on International Symposium on Symbolic and Algebraic Computation*, pages 267–274, 2019.

[BRT20] Paul Breiding, Kemal Rose, and Sascha Timme. Certifying zeros of polynomial systems using interval arithmetic. *arXiv preprint arXiv:2011.05000*, 2020.

[BT18] Paul Breiding and Sascha Timme. HomotopyContinuation.jl: A package for homotopy continuation in Julia. In *International Congress on Mathematical Software*, pages 458–465. Springer, 2018.

[CDLS] Justin Chen, Timothy Duff, Anton Leykin, and Mike Stillman. SLPexpressions.m2: a Macaulay2 package for straight line programs and expressions for evaluation circuits.

[CLL14] Tianran Chen, Tsung-Lin Lee, and Tien-Yien Li. Hom4PS-3: a parallel numerical solver for systems of polynomial equations based on polyhedral homotopy continuation methods. In *International Congress on Mathematical Software*, pages 183–190. Springer, 2014.

[GPV11] Elizabeth Gross, Sonja Petrović, and Jan Verschelde. PHCpack in Macaulay2. *arXiv preprint arXiv:1105.4881*, 2011.

[GS02] D. R. Grayson and M. E. Stillman. Macaulay2, a software system for research in algebraic geometry. Available at http://www.math.uiuc.edu/Macaulay2/, 2002.

[HS11] Jonathan D Hauenstein and Frank Sottile. alphaCertified: Software for certifying numerical solutions to polynomial equations. Available at http://math.tamu.edu/~sottile/research/stories/alphaCertified, 2011.

[HS12] J. D. Hauenstein and F. Sottile. Algorithm 921: alphaCertified: certifying solutions to polynomial systems. *ACM Transactions on Mathematical Software (TOMS)*, 38(4):28, 2012.

[Kra69] Rudolf Krawczyk. Newton-algorithmen zur bestimmung von nullstellen mit fehlerschranken. *Computing*, 4(3):187–201, 1969.

[Lan83] Serge Lang. *Real analysis*. Addison-Wesley Publishing Company, Advanced Book Program, Reading, MA, second edition, 1983.

[Lee19] Kisun Lee. Certifying approximate solutions to polynomial systems on Macaulay2. *ACM Communications in Computer Algebra*, 53(2):45–48, 2019.

[Ley11] A. Leykin. Numerical algebraic geometry. *Journal of Software for Algebra and Geometry*, 3(1):5–10, 2011.
[LVZ06] A. Leykin, J. Verschelde, and A. Zhao. Newton’s method with deflation for isolated singularities of polynomial systems. *Theoretical Computer Science*, 359(1-3):111–122, 2006.

[Moo77] R. E. Moore. A test for existence of solutions to nonlinear systems. *SIAM Journal on Numerical Analysis*, 14(4):pp. 611–615, 1977.

[Sma86] S. Smale. Newton’s method estimates from data at one point. *The Merging of Disciplines: New Directions in Pure, Applied, and Computational Mathematics*, 1986.

[SW05] A. Sommese and C. Wampler. *The Numerical Solution of Systems of Polynomials Arising in Engineering and Science*. World Scientific, 2005.

[Ver99] Jan Verschelde. Algorithm 795: PHCpack: A general-purpose solver for polynomial systems by homotopy continuation. *ACM Transactions on Mathematical Software (TOMS)*, 25(2):251–276, 1999.

Department of Mathematics, University of California San Diego, 9500 Gilman Drive, La Jolla, CA 92093

E-mail address: kil004@ucsd.edu

URL: https://klee669.github.io